

제정 기술표준원고시 제2001 - 30호 (2001. 2. 14)
개정 기술표준원고시 제2003 - 523호 (2003. 5. 24)

전기용품안전기준

K 60216-3-2

[KS C IEC 2002]

전기절연재료의 내열성 결정지침

제3부: 내열성 산출 지침

제2절: 불완전 데이터의 산출: 시험중간에서 종료까지의 내열특성

목 차

서문	2
서론	2
1. 적용범위 및 목적	2
2. 산출원리	11
3. 내열지수 유도 지침	17
4. 참고문헌	31
부속서 A	35
부속서 B	39
부속서 C	47
부속서 D	48
부속서 E	57
부속서 F	59

한국산업규격
전기절연재료의 내열성 결정지침
제3부: 내열성 산출 지침
제2절: 불완전 데이터의 산출: 시험중간에서
종료까지의 내열특성

KS C IEC
60216-3-2:2002
(IEC 60216-3-2:1993, IDT)

Guide for the determination of thermal endurance properties of electrical insulating materials

Part 3: Instructions for the calculating thermal endurance characteristics
Section 2 : Calculations for including the median time to end-point (equal test groups)

서문

본 규격은 1993년에 제1판으로 발행된 IEC 60216-3-2 (Guide for the determination of thermal endurance properties of electrical insulating materials Part 3: Instructions for the calculating thermal endurance characteristics Section 2: Calculations for including the median time to end-point (equal test groups))의 기술적 내용을 변경하지 않고 원문을 첨부하여 한국산업잠정규격으로 제정한 것이다.

서론

KSC IEC 60216은 다섯 개의 부로 구성되어 있다.

- 제1부: 노화절차 및 시험결과 평가의 지침(KSC IEC 60216-1)
- 제2부: 시험 기준의 선택(KSC IEC 60216-2)
- 제3부: 내열성 산출 지침(KSC IEC 60216-3)
- 제4부: 노화 오븐(KSC IEC 60216-4)
- 제5부: 내열성 적용 지침(KSC IEC 60216-5)

제3부는 정규 분포된 시험 결과의 평균값을 토대로 하는 네 개의 절로 구성되어 있다. 평균값 처리가 부적합한 경우 극한값 통계를 고려하기 위해 다른 통계 절차의 관련 절을 추가할 수도 있지만, 현재로서는 이러한 방법들을 내열 표준에 사용할 수 있을 정도로 경험이 충분하지 않다.

이 문서는 제2절에 해당한다. 제2절은 산출 지침만으로 구성되어 있다. 관련 통계 이론은 KSC IEC 493의 해당 부 및 명시한 참고문헌에서 찾을 수 있다.

새로운 제3부는 현재 준비중이며, 현재 1절과 2절과 함께 개별 절의 초안: 파괴 시험 절차에서 얻은 정규 분포된 결과에 대한 산출 절차를 개발하고 있다.

1. 적용범위 및 목적

KSC IEC 60216의 제3부는 KSC IEC 60216-1과 60216-2에 따라 얻은 데이터로부터 내열성을 산출할 수 있는 지침을 제공한다.

제1절은 수학적 배경 기준이 상세하게 작성된 KSC IEC 60493-1에서 설정한 통계 원리에 따라, 정규 분포된 완전

데이터에 대한 주요 순차 산출 계획을 제공한다.

제3절은 동일한 크기의 그룹에서 얻은 불완전 시험 결과에 대한 산출 순서를 제공한다. 실제 예로서, "BASIC"으로 작성된 컴퓨터 프로그램과 해당 통계표가 부속서에 제공하고 있다.

2. 산출 원리

2.1 통계 원리

제3절에 명시되어 있는 산출 절차와 지침은 KSC IEC 60493-1에서 설정한 원리와 가정을 토대로 한다. 가정은 다음과 같이 간단한 형태로 표현할 수 있다(KSC IEC 60493-1, 3.7.1절 참조).

- 1) 주어진 끝점에 도달하는데 걸리는 시간의 로그 평균("끝점 도달 평균시간")과 열역학 (절대) 노화 온도 사이의 관계는 선형적이다.
- 2) 이 선형 관계로부터 끝점 도달시간의 로그 편차값은 정규 분포되며, 분산은 노화 온도와 무관하다.

이용 가능한 데이터가 불완전한 경우, 각 그룹의 중앙값 이상에서 끝점 도달시간이 결정되지 않는다는 점에서, 각 그룹내의 끝점 도달시간의 로그 평균 및 분산의 불편 추정치와 평균 분산의 불편 추정치는 단순 대수 방법으로 결정할 수 있다(4절의 참조 문서 참조).

이러한 추정치는 KSC IEC 60216의 3부 1절에서 평균과 분산의 추정치에 대해 사용한 것과 유사한 방법으로 사용한다.

최선의 선형 관계 계수 추정치는 최소제곱법으로 얻는다. 그리고 이 선형 관계를 사용하는 추정치와 관련한 신뢰 한계치는 일반화된 회귀분석법을 사용하여 얻을 수 있다.

2.2 분석용 데이터

실험 데이터는 온도값(θ °C)과 끝점 도달시간(t 시간)으로 얻는다. 각 값은 x 나 y 값으로 변환된다.

$$x_i = \text{열역학 (절대) 온도의 역수의 값}(i) = 1 / (\theta_i + 273)$$

$$y_{ij} = i\text{번째 노화 그룹}(i)\text{에서 시간의 값}(j)\text{의 로그 } \log(t_{ij})^1$$

n_i = 그룹(i)의 y 값의 수

k = 그룹의 수

t_{ij} 의 개별값을 끝점에 도달하기 전에 마지막 보증 시험 사이클의 중간점으로 취한다.

끝점도달 m 번째 시간 t_{im} 까지 산출하는데 값이 포함된다.

$$m = (n + 1)/2 \quad n\text{이 홀수인 경우}$$

또는

$$m = n/2 + 1 \quad n\text{이 짝수인 경우}$$

2.3 제한사항

- a) 모든 그룹에서 초기 시료의 수는 동일하지만 11개 이상이다.

1) 원칙적으로, 선택이 시중 일관적이면, Napierian(로그 밑 e) 또는 Briggsian (로그 밑 10)을 사용하여 산출할 수 있다. 이러한 이유로, 기호 $\log(y)$ 가 3.1.5절을 제외한 모든 절에 사용된다.

일부 중간 결과를 KSC IEC 60216-3 (1980)과 216-4 (1980)의 2판과 비교할 수 있도록 하기 위해, 컴퓨터 프로그램 (부록 D)은 밑이 10인 로그를 사용한다.

- b) 각 그룹에서의 시험은 끝점 도달 중앙시간에 도달할 때까지 연속적이어야 한다.
- c) 각 그룹에 대한 사이클 시간은 끝점 도달 중앙시간이 8번째 사이클보다 더 일찍 도달하지 않고 12번째 보다 늦지 않도록 선택한다.
- d) 첫 번째 사이클 내의 끝점 도달시간은 기존의 알려져 있는 수학적 절차로 처리할 수 없으며, 무효한 것으로 취급해야 한다.

다음 중 하나를 적용한다.

- i) 새로운 그룹의 시료로 다시 시작한다.
- ii) 이 절차에 의해 도입된 수학적 부정확성이 작을 경우 부록 F의 절차를 채택한다.

끝점이 첫 번째 사이클 동안 하나 이상의 시료에 도달하는 경우, 이 그룹을 버리고 다른 그룹을 시험해야 하며 중요한 실험 기법에 각별한 주의를 기울인다.

2.4 통계적 검정

산출 절차에는 세 가지 시험이 포함된다.

- a) 분산 동일성(Bartlett의 χ^2) 검정법
- b) 선형성 (F) 검정
- c) 산포 (신뢰 구간) 검정법

검정 b)와 c)는 통계적으로 중요하지만 너무 작아 실제적인 중대한 결과를 갖지 못하는 이상적인 행동으로부터 편차를 검출할 수 있다. 이러한 환경을 충족시킬 수 있는 절차를 포함시킨다.

2.4.1 Bartlett χ^2 검정법

주요 불평등 이외의 분산의 불평등은 온도 지수나 신뢰 구간의 심각한 오류를 유도하는 것으로 간주하지 않는다. χ^2 값을 산출하고, 이 값이 0.05의 유효 레벨에 해당하는 것보다 크다면, 3.1.5절에 따라 χ^2 의 값과 유효 수준을 보고한다.

2.4.2 선형성을 위한 F 검정법

F 비는 회귀선(s_2^2 , 11식 참조)에서 노화 그룹(s_i², 10식 참조) 내 분산까지 노화 그룹 평균 편차의 분산 비이다. F값이 높으면 회귀선 편차값이 높거나 그룹 내 데이터 분산값이 낮을 수 있다.

F비는 0.05의 유효 레벨에서 표의 값 F_0 에 대해 검정된다.

매우 작은 산포 데이터의 경우, 비선형성이 실제로 거의 중요하지 않지만 통계적으로 중요한 것으로 검출될 수 있다.

F 검정에는 이 방법으로 실패하는 경우에도 어떤 결과를 얻을 수 있도록 다음과 같은 절차가 포함된다.

- a) F 검정이 수용가능한 결과를 제공할 수 있도록 노화 그룹(s_7^2) 내 분산값을 증가시킨다 즉, s_7^2 을 조정값

$(s_f^2)a = s_i^2(F/F_0)$ 으로 대체한다. 3.1.6절 b) 참조.

b) 이 조정값을 사용하여 결과의 하위 신뢰 한계 TC_a 를 산출한다.

c) 하위 신뢰 구간($TI - TC_a$)가 수용가능한 것으로 판명되면($TI - TC_a \leq 0.6$ HIC, 2.4.3절 참조), 비선형성은 실제로 중요하지 않은 것으로 인정한다.

2.4.3 데이터 산포 검정

온도 지수, 하위 신뢰 한계, 2등분 구간(각각 TI, TC, HIC)을 산출하였을 때, 하위 신뢰 구간($TI - TC$)은 다음의 경우에 수용 가능한 것으로 인정한다.

$$TI - TC \leq 0.6 \text{ HIC}$$

(또는 해당하는 경우 $TI - TC_a \leq 0.6 \text{ HIC}$)

하위 신뢰 구간이 작은 차이만큼 0.6 HIC를 초과하여 1.6 HIC 이하에 있는 경우, 조정된 TI_a 를 산출하여, 즉 산출된 TI에서 하위 신뢰 구간이 0.6 HIC를 초과한 양을 빼어 산출하여 ($F \leq F_0$ 라고 가정) 사용 가능한 결과를 얻을 수 있다.

즉, $0.6 \text{ HIC} < TI - TC \leq 1.6 \text{ HIC}$ 이고, 동시에 $F \leq F_0$ 이면 TI으로 회귀식(3.1.8절 참조)에서 산출한 TI값 대신 아래 값을 보고한다.

$$TI_a = TI - [(TI - TC) - 0.6 \text{ HIC}] = TC + 0.6 \text{ HIC}$$

위의 조건이 충족되지 않으면, 시험 결과를 3.2.3절과 같이(KSC IEC 60216-1, 12.5절 참조) $TI_g = \dots$, $HIC_g = \dots$ 의 형태로만 보고할 수 있다.

2.5 산출의 내부 정확도

산출 단계의 대부분에는 숫자와 비교하여 차이가 작은 경우 수의 차의 합 또는 이러한 차이의 제곱의 합이 포함된다. 이러한 환경에서, 3자리 결과의 정확도를 갖기 위해 최소한 6자리 유효 숫자의 내부 정확도로 산출해야 한다. 산출이 반복적이고 지루하기 때문에, 프로그래머블 계산기나 마이크로컴퓨터를 사용하여 산출할 것을 강력히 권장한다. 컴퓨터를 사용하면 10이나 12자리의 내부 정확도를 쉽게 얻을 수 있다.

3. 내열 특성 유도 지침

3.1 수치 유도와 내열 그래프

3.1.1 그룹 평균, 분산, 평균 분산의 추정

(이 지침은 끝점 도달시간이 최대 중앙시간으로 알려진 동일한 그룹의 경우에 적용한다.)

i) 증가 순서대로 각 그룹에서 끝점 도달시간의 로그를 배열한다.

$$y_{i1} \leq y_{i2} \leq \dots \leq y_{im}$$

ii) 평균 추정치를 산출한다.

$$\bar{y}_i = (1 - \mu)y_{im} + \frac{\mu}{(m-1)} \sum_{j=1}^{m-1} y_{ij} \quad (1)$$

iii) 분산 추정치를 산출한다.

$$s_{ij}^2 = \alpha \sum_{j=1}^{m-1} (y_{im} y_{ij})^2 + \beta \left[\sum_{j=1}^{m-1} (y_{im} - y_{ij}) \right]^2 \quad (2)$$

계수 μ , α , β (그리고 식 11에서 사용한 ϵ)는 부록 E에 명시되어 있다.

3.1.2 일반 평균

x , y , N 값을 산출한다. 3절의 모든 나머지 합은 1 ~ k 의 범위에서 지수 I에 대한 것이다.

$$N = mk \quad (3)$$

$$\bar{x} = \sum x_j / k \quad (4)$$

$$\bar{y} = \sum y_j / k \quad (5)$$

3.1.3 회귀식과 내열 그래프

회귀식의 계수를 산출한다.

$$y = a + bx \quad (6)$$

여기서

$$b = \frac{\sum x_j \bar{y}_j - k \bar{x} \bar{y}}{\sum x_j^2 - k \bar{x}^2} \quad (7)$$

$$a = \bar{y} - b \bar{x} \quad (8)$$

회귀선이 확정되면, $y = \log(t)$ 를 세로축에 $x = 1/(\theta + 273)$ 을 가로축으로 하여 내열 그래프에 회귀선을 그린다. 대개 x 는 오른쪽에서 왼쪽으로 증가하는 것으로 그리고, °C에서의 θ 해당값을 이 축에 표시한다(그림 1 참조).

- 비고 - 특수 그래프 종이를 사용할 수 있다.

또 다른 방법으로, 해당 비선형 눈금에 그래프를 그리기 위해 이 산출을 실시한 컴퓨터 프로그램에 서브루틴을 포함시킬 수 있다(부록 D 참조, 부록 D에 서브루틴이 포함되어 있다).

3.1.1절에서 얻은 개별값 $y_{ij} = \log(t_{ij})$ 과 평균값 y_i 를 아래 해당값에서 그래프에 그린다.

$$x_j = \frac{1}{\theta_j + 273} \quad (9)$$

하위 95% 신뢰선을 그려 내열 그래프를 완성한다(3.1.7절 참조).

3.1.4 분산 산출

노화 그룹 내의 분산 가중평균값을 산출한다.

$$s_{I'}^2 = \frac{\sum s_{ij}^2}{k} \quad (10)$$

회귀선으로부터 노화 그룹 평균의 편차의 가중 분산을 산출한다.

$$s_2^2 = \frac{1}{\varepsilon} \frac{\sum (\bar{y} - Y_i)^2}{k-2} \quad (11)$$

여기에서, ε 은 부록 E에 명시되어 있다.

$$Y_i = a + bX_i \quad (12)$$

이는 다음과 같이 단순화할 수 있다.

$$s_2^2 = \frac{(\sum y_i^2 - k\bar{y}^2) - b(\sum x_i y_i - k\bar{x}\bar{y})}{\varepsilon(k-2)} \quad (13)$$

x값의 중앙 두 번째 모멘트를 산출한다.

$$\mu_2(x) = \frac{\sum x_i^2 k \bar{x}^2}{k} \quad (14)$$

3.1.5 분산의 평등 검정

Bartlett 방법으로 χ^2 값을 산출한다.

$$\chi^2 = \frac{\ln a}{c} [(N-k) \log_q s_i^2 - \sum (m-1) \log_q s_{ij}^2] \quad (15)$$

여기서
$$c = 1 + \frac{\sum (\frac{1}{m-1}) \frac{1}{N-k}}{3(k-1)} \quad (15a)$$

그리고 \log_q 는 밑이 q인 로그를 나타낸다. Briggsian 로그(밑이 10인 로그)를 사용하면, $\ln q \approx 2.3$, Napierian 로그(밑이 e인 로그)를 사용하면 $\ln q = 1$ 이다.

자유도 $f = k-1$ 에 대해 χ^2 의 값을 표의 값 χ_0^2 과 비교한다(부속서 C, 표 C1 참조).

χ^2 값이 0.05의 유효 수준에 대한 표의 값보다 크다면, χ^2 값과 χ^2 보다 낮은 최고값에 대한 표의 유효 수준을 보고한다. 또 다른 방법으로, χ^2 과 그 유효 수준을 컴퓨터 프로그램(예 ; 부속서 D의 프로그램)으로 산출할 경우 이를 보고한다.

3.1.6 선형성 검정(F-검정)

0.05의 유효 수준에서 F 비 검정으로 회귀선 s_2^2 의 편차 분산을 k 측정 집합 내 합동 분산 s_1^2 와 비교한다.

비를 산출한다.

$$F = s_2^2 / s_1^2 \quad (16)$$

그리고 이 값을 $f_n = k-2$ 와 $f_d = N-k$ 자유도로 표의 값 F_0 와 비교한다(부속서 C 표 C2 참조).

$$F_0 = F(0.95, K-2, N-k)$$

a) $F \leq F_0$ 이면, 합동 분산 추정치를 산출한다.

$$s^2 = \frac{(k-2)s_2^2 + (N-k)s_1^2}{N-2} \quad (17a)$$

b) $F > F_0$ 이면, s_1^2 을 $(s_1^2)_a = s_1^2(F/F_0)$ 로 조정하고 조정된 값 s_a^2 을 산출한다.

$$s_a^2 = \frac{(k-2)s_2^2 + (N-k)(s_1^2)_a}{N-2} \quad (17b)$$

3.1.7 X와 Y 추정치의 신뢰 한계치

a) Y-추정치

주어진 X값에 해당하는 추정값 Y의 하위 95% 신뢰 한계는 다음과 같다.

$$Y_c = Y - t s_Y \quad (18)$$

여기서 $Y = a + bX$,

$$s_Y^2 = \frac{\varepsilon s^2}{k} \left(1 + \frac{(X - \bar{X})^2}{\mu_2(x)} \right) \quad F \leq F_0 \text{일 경우, 위의 수식. (19a)}$$

$$s_Y^2 = \frac{\varepsilon s_a^2}{k} \left(1 + \frac{(X - \bar{X})^2}{\mu_2(x)} \right) \quad F > F_0 \text{일 경우, 아래의 수식. (19b)}$$

그리고 t는 95%의 신뢰 수준에서 자유도 N-2인 Student의 t, $t(0.95, N-2)$ 의 표의 값이다(부속서 C, 표 C3 참조).

내열 그래프의 신뢰 한계 곡선(3.1.3절)의 경우, Y_c 는 해당 범위에서 몇 가지 (X, Y)값의 쌍에 대해 산출하고, 이 점(X, Y_c)을 통해 그려진 곡선을 그래프상에 좌표로 나타낸다.

b) X-추정치

끝점 도달시간에 해당하는 X값의 상위 95% 신뢰 한계를 $F \leq F_0$ 인 경우에 다음 식으로 산출한다.

$$X_c = \bar{x} + \frac{(Y - y)}{b_r} + \frac{t s_r}{b_r} \quad (20)$$

여기서

$$Y = \log(t_f) \quad (20a)$$

$$b_r = b - \frac{\epsilon t^2 s^2}{k b \mu_2(x)} \quad (20b)$$

$$s_r^2 = \frac{\epsilon s^2}{k} \left(\frac{b_r}{b} + \frac{(X - \bar{x})^2}{\mu_2(x)} \right) \quad (20c)$$

$$X = \frac{Y - a}{b} \quad (20d)$$

$F > F_0$ 이면 s^2 대신 s_a^2 값을 (20b)와 (20c)식에 삽입한다.

t 는 95%의 신뢰 수준에서 자유도가 $N-2$, 즉 $t(0.05 - N-2)$ 인 표의 Student의 t 값이다(부록 C, 표C3)

어떤 온도 추정치의 하위 95% 신뢰 한계는 해당하는 X 추정치, X_c 의 상위 신뢰 한계에서 산출할 수 있다.

$$\vartheta_c = \frac{1}{X_c} - 273 \quad (21)$$

3.1.8 내열성 특성의 산출

3.1.3절의 회귀식(6)에서 20 kh의 끝점 도달시간에 해당하는 온도 $^{\circ}\text{C}$ 를 산출한다. 이 값이 온도 지수 TI 이다.

10 kh의 끝점 도달시간, TI_{10} 에 해당하는 온도를 같은 방법으로 산출한다. 2등분 구간, HIC는 $TI_{10} - TI$ 이다.

조정값 s_a^2 을 사용하는 경우 $TI(Y = \log[20,000])$, TC , TC_a 의 하위 95% 신뢰 한계를 3.1.6 b)의 방법으로 산출한다.

$(TI - TC)/HIC$, $(TI - TC_a)/HIC$ 값을 결정한다.

이 값이 0.6 이하일 경우, 시험 결과를 3.1.10절에서 언급한 대로 $TI(HIC)$ 형식으로 보고한다(KSC IEC 60216-1, 14 절 참조).

$0.6 < (TI - TC)/HIC \leq 1.6$ 이고, $F \leq F_0$ (3.1.5절)이면, HIC와 함께 $TI_a = TC + 0.6 HIC$ 값을 $TI(HIC)$ 형식으로 보고한다.

모든 다른 경우에, 결과는 3.2.3절에 명시한 대로 $TI_g = \dots$, $HIC_g = \dots$, 형식으로 보고해야 한다.

3.1.9 통계적 검정 조사 및 보고

표 I에서, “검정”으로 시작하는 열의 조건이 충족되지 않으면, 최종 열에 나타난 대로 조치한다. 조건이 충족되면, 다음 단계에서 조치가 나타난다. 이 순서는 또한 내열 산출을 위한 결정 흐름도(그림 1)에서도 나타나 있다.

표 I

단계	검정(?) 또는 조치	참조	No인 경우 조치
1	? 최장 끝점도달 평균시간 > 5000시간		12 단계로 이동
2	?외삽* ≤ 25 K		12 단계로 이동
3	? $P(\chi^2, f) \leq 0.05$	3.1.5	χ^2 과 P를 보고하고 4단계로 이동
4	? $F \leq F_0$	3.1.6	s12 - s12 F/F0를조정하고 9단계로 이동
5	? $TI - TC \leq 0.6(\text{HIC})$	3.1.8	7단계로 이동
6	TI(HIC)를 보고		
7	? $TI - TC \leq 1.6(\text{HIC})$	3.1.8	11단계로 이동
8	TIa = TC + 0.6(HIC)와 HIC를 TI(HIC) 형식으로 보고	2.4.3	
9	? $TI - TC_a \leq 0.6(\text{HIC})$	2.4.2	11단계로 이동
10	TI(HIC)를 보고		
11	TI _g 와 HIC _g 를 보고		
12	하위 온도에서 새로운 그룹 시험		

* 외삽 = 최저 시험 온도 - TI

3.1.10 보고

필요한 형식과 보고 내용이 KSC IEC 60216-1 14절에 명시되어 있다. 그러나 단일 수치 결과 TI(HIC)가 전체 시험 데이터가 양질한 것인지를 나타낼 수 없고 데이터의 평가를 완료할 수 없기 때문에 보고의 부분으로 내열 그래프를 제출해야 한다.

3.2 그래프 유도

3.2.1 완전한 그래프 유도

각 시험 온도(2.2절 참조)에서 끝점 도달시간의 최고값을 시험 데이터에서 유도한다. 이 값과 시험 온도 ??의 값을 3.1.3절에서 언급한 특수 그래프 종이에 좌표로 나타낸다. 내열 그래프를 그리고, 시각적인 방법으로 “최적의 라인”을 얻고, 온도 지수 TI_g를 끝점 도달시간이 20,000시간인 온도로 평가하고, 2등분 구간 HIC_g를 온도차 TI10 - TI_g로 평가한다. 여기에서 T₁₀은 끝점 도달시간이 10,000시간인 온도이다.

3.2.2 부분적 그래프 유도

3.1.3절과 같이 회귀식의 a와 b 계수를 산출하고, 이를 이용하여 내열 그래프를 그린다. 동일한 그래프에, 각 시험 온도에서 끝점도달 중앙 시간의 최고값을 표시한다(2.2절 참조).

내열 그래프로부터, 3.1.8절과 같이 TI_g와 HIC_g를 결정한다.

3.2.3 보고

“TI_g = , HIC_g = .”를 보고한다.

4. 참조 문서

J.G. Saw, *Biometrika*, 46, 150(1959); "Estimation of the Normal Population Parameters given a Singly Censored Sample".




그림 1. - 내연 산출을 위한 결정 흐름도

부속서 A

기호 목록

기호		절
a	회귀 계수	3.1.3
b	회귀 계수	3.1.3
b_{τ}	중간 상수	3.1.7b)
c	중간 상수	3.1.5
f	자유도의 수	
F	Fisher 분포 확률 변수	3.1.6
F_0	F 표의 값	3.1.6
HIC	2등분 구간	3.1.8
HIC_g	TI_g 에 해당하는 2등분 구간	3.2
i	노출 온도의 차수 번호	2.2
j	끝점 도달시간의 차수 번호	2.2
k	노출 온도의 수	2.2
$\log(x)$	밑이 규정되지 않은 x의 로그	2.2
m	산출에 포함된 끝점도달 최종 시간의 차수	2.2
n	각 시험 그룹에서 시료의 수	2.2
N	시험한 시료의 총 수	3.1.2
P	χ^2 의 유효 수준	표 C1
S^2	합동분산추정치	3.1.6 a
S_a^2	s^2 의 조정값	3.1.6 b)
S_r^2	중간 상수	3.1.7 b)
S_Y^2	Y의 분산	3.1.7 a)
S_j^2	S_{ii}^2 의 가중평균	3.1.4
$(s_i^2)_a$	s_j^2 의 조정값	3.1.6 b)
s_{ii}^2	θ_i 에서 노출된 시료 집합 내의 분산	3.1.1
S_2^2	회귀선에 대한 분산	3.1.4
t	Student 분포 확률 변수	3.1.7 a)
t_{τ}	추정을 위해 선택한 끝점도달시간의 값	3.1.7 b)
t_{ij}	θ_i 에서 노출된 j번째 시료의 끝점도달시간	2.2
tim	산출에 포함된 끝점도달 최고 시간	2.2
TC	TI의 하위 단측 95% 신뢰 한계	3.1.8
TC_a	TC의 조정값	3.1.8
TI	온도 지수	3.1.8
TI_a	TI의 조정값	2.4.3
TI_g	그래픽 수단이나 정의된 신뢰 한계치없이 얻은 온도 지수	3.2
TI_{10}	10 kh에서의 온도 지수	3.1.8

x	독립 변수: 열역학 (절대) 온도의 역수	2.2
X	회귀식에서 규정된 y 값에서의 x 값	3.1.7 b)
X_c	X 에 대한 상위 단측 95% 신뢰 한계	3.1.7 b)
x_i	θ_i 에 해당하는 열역학 온도의 역수	2.2
\bar{x}	x_i 의 가중 평균	3.1.1
y	종속 변수: 끝점도달시간의 로그	2.2
Y	회귀식으로부터 규정된 x 값에서의 y 값	3.1.7 b)
Y_c	Y 에 대한 하위 단측 95% 신뢰 한계	3.1.7 a)
Y_i	$x = x_i$ 에서 회귀식의 y 값	3.1.4
y_{ij}	t_{ij} 에 해당하는 y 값	2.2
\bar{y}_i	θ_i 에서 y_{ij} 값의 평균	3.1.1
\bar{y}_i	\bar{y}_i 의 가중평균	3.1.2
α	불완전 데이터의 추정 계수	3.1.1
β	불완전 데이터의 추정 계수	3.1.1
ϵ	불완전 데이터에 대한 평균 추정치의 분산 추정치 계수	3.1.4
??	온도(단위: °C)	2.2
??	X_c 에 해당하는 온도에서 하위 단측 95% 신뢰 한계	3.1.7 b)
??	노출 온도(단위: °C)	2.2
μ	불완전 데이터에 대한 평균 추정 계수	3.1.1
$\mu_2(x)$	x 값의 중앙 두 번째 모멘트	3.1.4
χ^2	χ^2 분포 확률 변수	3.1.5
χ_0^2	χ^2 의 표의 값	3.1.5

부속서 B

실례

모든 값은 t_{ij} 값의 첫 번째 표와 그 로그를 제외하고 값8자리 유효숫자로 보고한다. 그러나, 여기에서 보고된 산출 값은 14자리의 내부 정확도로 실시하였고 이 정확도까지 초기에 유도하였다. 이는 필요한 엄격도 보다 훨씬 높은 것이다.

각 그룹의 시료의 수 : 21

260°C에서 시험된 그룹은 첫 번째 사이클 동안 불합격이었고, 따라서 12번째 불합격까지 계속되었다. 20개중에서 11 개를 분석하였다.

노출온도	240		260		280			
x_i 의 값	$1,94932 \cdot 10^{-3}$		$1,87617 \cdot 10^{-3}$		$1,80832 \cdot 10^{-3}$			
시료의 개수	j	t	y	t	y	t	y	
	1	1 044	3,01870	-	-	336	2,52633	
	2	1 044	3,01870	1 368	3,13609	336	2,52633	
	3	3 144	3,49748	1 368	3,13609	432	2,63548	
	4	3 144	3,49748	1 368	3,13609	432	2,63548	
	5	4 548	3,65782	1 896	3,27784	624	2,79518	
	6	4 548	3,65782	2 136	3,32960	624	2,79518	
	7	4 548	3,65782	2 136	3,32960	720	2,85733	
	8	5 136	3,71062	2 136	3,32960	720	2,85733	
	9	6 888	3,83809	2 136	3,32960	720	2,85733	
	10	6 888	3,83809	2 136	3,32960	816	2,91169	
	11	8 316	3,91991	2 832	3,45209	912	2,95999	
	12	-	-	2 832	3,45209	-	-	
$\Sigma y_{ij}, j=1 \dots m-1$			35,392638		32,786196		27,397702	방정식 1
$\Sigma (y_{im} - y_{ij})^2, j=1 \dots m-1$			2,2445421		0,40496778		0,67500617	방정식 2
\bar{y}_i			3,9199144		3,4391277		2,9599948	방정식 1
s_{li}^2			0,22256300		0,036344813		0,064599547	방정식 2
μ	부록 E		0		0,07474052		0	방정식 1
α	부록 E		0,125180504		0,123720203		0,125180504	방정식 2
$\beta \times 10^3$	부록 E		-4,10278708		-4,57177272		-4,10278708	방정식 2
$\epsilon \times 10^2$	부록 E		7,32597474		(7,283429)		7,32597474	방정식 11

약간 초과인 산포, 수정됨

방정식	량	값	
3	N	33	
4	\bar{x}	$1,8779362 \cdot 10^{-3}$	$(\sum x_i = 5,6338086 \cdot 10^{-3})$
5	\bar{y}	3,4396790	$(\sum y_i = 10,319037)$
7	$\sum x_i \bar{y}_i$	$1,9446169 \cdot 10^{-2}$	
	$\sum x_i^2$	$1,0589878 \cdot 10^{-5}$	
	b	6 804,9189	
8	a	-9,3395246	
10	s_1^2	0,10749000	
11	s_2^2	$2,6855295 \cdot 10^{-3}$	
14	$\mu_2(x)$	$3,3150304 \cdot 10^{-9}$	
15a	c	1,0398990	
15	χ^2	8,3700067	
16	F	0,02498393	
17a	s^2	0,10410921	
18	t	1,6930490	(내부적으로 산출)
20d	$X_{20\ 000}$	$2,0045139 \cdot 10^{-3}$	
20	X_c	$2,0342146 \cdot 10^{-3}$	
20a	$Y_{20\ 000}$	4,3010300	
20b	b_r	6781,2528	
20c	s_r^2	0,057072991	
	$\vartheta_{20\ 000}$	225,87407	(TI)
21	ϑ_c	218,59021	(TC)
	$\vartheta_{10\ 000}$	237,13204	
	HIC	11,257977	
	0,6 HIC	7,0666924	
	$\vartheta_c + 0,6 \text{ HIC}$	225,65691	(산포도 수정)

TI (HIC): 225,3 (11,3)

신뢰한계 = 218,6

$F(n, d) = 0,025 (1, 30)$

$F_0 = 4,2$

$\chi^2(f) = 8,4 (2)$

확률 = 0,0152

온도	평균시간	분산 (log)
240	8 316,0	0,222
260	2 748,7	0,0363
280	912,0	0,0646
합동분산 (s_1^2)		0,107

온도 240

시간 :

1044,0 1044,0 3144,0 3144,0 4548,0 4548,0 4548,0 5136,0 6888,0 6888,0 8316,0

온도 260


시간 :

1368,0 1368,0 1368,0 1896,0 2136,0 2136,0 2136,0 2136,0 2136,0 2832,0 2832,0

온도 280

시간 :

336,0 336,0 432,0 432,0 624,0 624,0 720,0 720,0 720,0 816,0 912,0



_____ 회귀선
 - - - - - 분점 도달시간의 하위 95% 신뢰 한계

그림 B1 - 이 내열 그래프는 부록 B의 실례의 데이터로부터 유도하였다.

부속서 C

χ^2 , t 및 F 분포의 fractiles

표 C1

		χ^2			
		P	0.050	0.010	0.005
f					
1		3.8	6.6	7.9	10.8
2		6.0	9.2	10.6	13.8
3		7.8	11.3	12.8	16.3
4		9.5	13.3	14.9	18.5
5		11.1	15.1	16.7	20.5
6		12.6	16.8	18.5	22.5

표 C2

		F				
		f_n	1	2	3	4
f_d						
12		4.8	3.9	3.5	3.3	3.1
13		4.7	3.8	3.4	3.2	3.0
14		4.6	3.7	3.3	3.1	3.0
15		4.5	3.7	3.3	3.1	2.9
16		4.5	3.6	3.2	3.0	2.9
17		4.5	3.6	3.2	3.0	2.8
18		4.4	3.6	3.2	2.9	2.8
19		4.4	3.5	3.1	2.9	2.7
20		4.4	3.5	3.1	2.9	2.7
25		4.2	3.4	3.0	2.8	2.6
30		4.2	3.3	2.9	2.7	2.5
40		4.1	3.2	2.8	2.6	2.5
50		4.0	3.2	2.8	2.6	2.4
100		3.9	3.1	2.7	2.5	2.3
500		3.9	3.0	2.6	2.4	2.2

표 C3

f	t
1	6.31
2	2.92
3	2.35
4	2.13
5	2.02
6	1.94
7	1.90
8	1.86
9	1.83
10	1.81
11	1.80
12	1.78
13	1.77
14	1.76
15	1.75
16	1.75
17	1.74
18	1.73
19	1.73
20	1.73
25	1.71
30	1.70
40	1.68
50	1.68
100	1.66
500	1.65

부속서 D

불완전 데이터에 대한 기본 프로그램

이 프로그램은 KSC IEC 60216-3-1의 프로그램과 동일한 BASIC 언어로 작성되어 있다.

프로그램의 목적은 세 가지 이상의 온도에서 노화된 동일한 크기의 시료 그룹에서 얻은 데이터를 분석하는 것이다. 노화는 각 그룹에서 중앙 불합격 후 중단되었다.

프로그램이 저장한 이름은 중요하지 않다. 왜냐하면 현재로서는 다른 내열 분석 프로그램과 직접 인터랙트할 의도가 없기 때문이다.

```
10 SCREEN 0:CLS:PRINT TAB(35) "IEC 216-3-3"
20 PRINT:PRINT"This program is for analysis of thermal endurance data obtained
  by cyclic"
30 PRINT"proof tests up to the median failure. The sizes of the temperature gr
  ups"
40 PRINT"must be the same, but the case of a failure in the first cycle, which
  must"
50 PRINT"be omitted from the calculations, is catered for."
60 DATA 640,200
70 KEY OFF:READ WD%,HT%
80 PRINT:PRINT:PRINT"Input may be from an existing disc file (D) or by direct
  entry from the"
90 PRINT"keyboard (K). Please press either K or D."
100 DEFDBL G,H,S,T,X,Y,P:DEFINT I-N
110 DEF FNTI (TIME)=FNTEMPERATURE (FNX2 (LOG (TIME) /LOG (10)))
120 DEF FNYC (X)=A#+B#*X-FNST (M*K) *SQR (S* (1+K* (X-SX/K) ^2/SB) *E#)
130 DEF FNX2 (Y)=(Y-A#) /B#
140 DEF FNX1 (TEMPERATURE)=1/ (TEMPERATURE+T0)
150 DEF FNTEMPERATURE (X)=1/X-T0
160 DEF FNSR (TIME)=SQR (S* (BR#/B#+ (FNX2 (LOG (TIME) /LOG (10)) -SX/K) ^2/ (SB/K)) *E#)
170 DEF FNXC (TIME)=SX/K+ (LOG (TIME) /LOG (10) -SY/K) /BR#+FNST (N) *FNSR (TIME) /BR#
180 DEF FNST (N)=1.6282+.0001688* (N-2) +1.8481/ (N-2)
190 DEF FNY (X)=A#+B#*X
200 DEF FNFO (N1,N2)=1.9993/N1+10.468/N2+1.8718
210 DEF FNU$ (A$)=CHR$ (ASC (A$) AND 223)
220 DEF FNA$ (A$)=CHR$ (((A$=CHR$ (0)+CHR$ (72)) AND 30) OR ((A$=CHR$ (0)+CHR$ (80))
  AND 31) OR ((A$=CHR$ (0)+CHR$ (77)) AND 28) OR ((A$=CHR$ (0)+CHR$ (75)) AND 29)
  OR (A$=CHR$ (13)) AND 13)
```

```

230 ENTRY$="":WHILE ENTRY$<>"D" AND ENTRY$<>"K":ENTRY$="":WHILE ENTRY$="":ENTRY$
=INKEY$:WEND:ENTRY$=FNU$(ENTRY$):WEND
240 IF ENTRY$="D" THEN 1890
250 'Keyboard entry
260 '*****
270 CLS:INPUT"Enter the required number of test temperatures ",K
280 INPUT"Enter the number of specimens in a group ",L
290 IF L MOD 2 THEN M=(L+1)/2 ELSE M=L/2+1
300 DIM N(K), TEMPERATURE(K), X(K), S1(K), SY(K), SSY(K), TIME(K,M), Y(K,M),
B#(5),N1(3),GA(3),GB(3),GM(3),GE(3),MEAN#(K)
310 FOR I=1 TO K
320 PRINT"Enter temperature";I:INPUT" ",TEMPERATURE(I)
330 PRINT"First cycle failure Y/N?";A$="":WHILE A$<>"Y" AND A$<>"N":A$="":WHILE
A$="":A$=INKEY$:WEND:A$=FNU$(A$):WEND
340 PRINT " ";A$:IF A$="Y" THEN N(I)=L-1 ELSE N(I)=L
350 FOR J=1 TO M
360 PRINT"Enter time to failure for sample";J;" if same as last, press RETURN";
370 INPUT" ",TIME(I,J):IF TIME(I,J)=0 THEN TIME(I,J)=TIME(I,J-1)
380 NEXT:NEXT
390 INPUT"Enter the filename for data saving ",FILENAME$
400 'Data review for editing
410 '*****
420 FOR I=1 TO K
430 IF N(I)<L THEN FIRST$=" Y" ELSE FIRST$=" N"
440 CLS:PRINT"Temperature";TAB(26) TEMPERATURE(I)
450 PRINT"First cycle failure";TAB(20) FIRST$
460 PRINT"Times";:FOR J=1 TO M
470 IF J>8 THEN X=41:Y=J-6 ELSE X=26:Y=J+2
480 LOCATE Y,X:PRINT TIME(I,J)
490 NEXT
500 LOCATE 22,1:PRINT"Do you wish to correct any value, Y/N ?"
510 A$="":WHILE A$="":A$=INKEY$:WEND:A$=FNU$(A$):IF A$="Y" THEN 1650
520 NEXT
530 CLS
540 'Primary calculation
550 '*****
560 RESTORE 3410
570 M1=0:WHILE M1<>M:READ M1:IF M1<>M THEN READ N1,Z,Z,Z,Z:WEND
580 FOR I=1 TO 3:READ N1(I),GA(I),GB(I),GM(I),GE(I)
590 READ J
600 NEXT
610 PRINT"Calculating ";FILENAME$:TO=273:N=K:FOR I=1 TO K:X(I)=FNX1(TEMPERATURE
(I))
620 FOR J=1 TO M:Y(I,J)=LOG(TIME(I,J))/LOG(10)
630 NEXT:NEXT
640 IF L MOD 2 THEN E#=GE(3) ELSE E#=GE(2)
650 FOR I=1 TO K
660 FOR J=1 TO 3
670 IF N1(J)=N(I) THEN A#=GA(J):B#=GB(J):M#=GM(J)
680 NEXT
690 FOR J=1 TO M-1
700 SY(I)=SY(I)+Y(I,J):SSY(I)=SSY(I)+(Y(I,M)-Y(I,J))^2
710 NEXT
720 MEAN#(I)=(1-M#)*Y(I,M)+M#*SY(I)/(M-1)
730 S1(I)=A#*SSY(I)+B#*((M-1)*Y(I,M)-SY(I))^2
740 SX=SX+X(I):SSX=SSX+X(I)^2:SPXY=SPXY+X(I)*MEAN#(I)
750 SY=SY+MEAN#(I):SSY=SSY+MEAN#(I)^2
760 NEXT
770 B#=(SPXY-SY*SX/K)/(SSX-SX^2/K)
780 A#=(SY-B#*SX)/K
790 M1#=CDBL(M)
800 FOR I=1 TO K

```

```

810 S1=S1+S1(I)
820 G=G+1/M1#
830 H=H+(M1#-1)*LOG(S1(I))
840 NEXT
850 S1=S1/K
860 CHI.SQUARED#=(K*(M-1)*LOG(S1)-H)/(1+(G-1/K/(M-1))/3/(K-1))
870 GOSUB 2690
880 S2=((SSY-SY^2/K)-B#*(SPXY-SX*SY/K))/E#/(K-2):F=S2/S1:F0=FNFO(K-2,M*K-K)
890 SB=SSX-SX^2/N
900 IF F>F0 THEN FLAG(3)=-1:F1=F/F0 ELSE FLAG(3)=0:F1=1
910 S=E#*((M-1)*K*S1*F1+(K-2)*S2)/(M*K-2)
920 BR#=B#-E#*FNST(M*K)^2*S/B#/SB
930 TIME=20000
940 HIGH.X=0:FOR I=1 TO K
950 IF X(I)>HIGH.X THEN HIGH.X=X(I):LOWEST=I
960 NEXT
970 GOSUB 1080 'for flag setting
980 Y1=5:Y2=1
990 X1=FNX1(10*INT(TI/10)-20):X2=X1-.0008
1000 SCALESTART=CINT(10*INT(TI/10)-10)
1010 SCALE.END=CINT(10*(INT(FNTEMPERATURE(X2)/10)))
1020 IF HIC>15 THEN INTERVAL=20 ELSE INTERVAL=10
1030 GOSUB 1200 'short report
1040 GOTO 2420 'for menu of further actions
1050 END
1060 'Statistical tests
1070 '*****
1080 TI=FNTE(TIME):TC=FNTEMPERATURE(FNXC(TIME))
1090 HIC=FNTE(TIME/2)-FNTE(TIME)
1100 EXTRAPOLATION=FNTEMPERATURE(HIGH.X)-TI
1110 IF EXTRAPOLATION>25 THEN FLAG(1)=-1 ELSE FLAG(1)=0
1120 LONGEST.MEAN.TIME=10^(MEAN#(LOWEST))
1130 IF LONGEST.MEAN.TIME<TIME/4 THEN FLAG(2)=-1 ELSE FLAG(2)=0
1140 DISPERSION=(TI-TC)/HIC
1150 IF DISPERSION>1.6 THEN FLAG(5)=-1 ELSE FLAG(5)=0
1160 IF DISPERSION>.6 THEN FLAG(4)=-1 ELSE FLAG(4)=0
1170 RETURN
1180 'report results according to statistical tests
1190 '*****
1200 CLS:IF FLAG(1) OR FLAG(2) OR FLAG(5) OR (FLAG(3) AND FLAG(4)) THEN GOSUB
1260 :RETURN
1210 IF FLAG(3) THEN PRINT"Minor non-linearity":PRINT
1220 IF FLAG(4) THEN PRINT"Slightly excessive dispersion, corrected":PRINT:RESULT
=TC+.6*HIC ELSE RESULT=TI
1230 PRINT"TI (HIC) : ";:PRINT USING "###.#";RESULT;:PRINT"(:";:PRINT USING"###.#"
;HIC;:PRINT")":PRINT:PRINT
1240 PRINT"Confidence limit = ";:PRINT USING "###.#";TC
1250 RETURN
1260 PRINT"Full TI cannot be reported.":PRINT
1270 IF FLAG(1) THEN PRINT"Extrapolation = ";:PRINT USING "###.#";EXTRAPOLA-
TION:PRINT
1280 IF FLAG(2) THEN PRINT"Longest mean time to end point = ";:PRINT USING "####"
;LONGEST.MEAN.TIME:PRINT
1290 IF FLAG(3) THEN PRINT"Failed F-test : F = ";:PRINT USING "####.#";F;:PRINT"
, F0 = ";:PRINT USING "###.#";F0:PRINT
1300 IF FLAG(5) THEN PRINT"Excessive dispersion.":PRINT
1310 PRINT:PRINT"TI = ";:PRINT USING"###.#";TI:PRINT"HIC = ";:PRINT USING "###.#"
;HIC
1320 RETURN
1330 'graphics - thermal endurance graph
1340 '*****
1350 SCREEN 2:XRANGE=X1-X2:YRANGE=Y2-Y1

```

```

1360 LOCATE 1,1,0:CLS:VIEW(WD%/100,HT%/100)-(WD%*.99,HT%*.99),0,1
1370 WINDOW(X2,Y1)-(X1,Y2)
1380 FOR TEMPERATURE%=SCALESTART TO SCALE.END STEP INTERVAL
1390 X=X1+X2-FNX1(TEMPERATURE%)
1400 IF TEMPERATURE% MOD 50=0 AND X-X2>XRANGE/40 THEN LOCATE 24,(X-X2)/XRANGE*80
-1:IF X1-X>XRANGE/20 THEN PRINT TEMPERATURE%;
1410 IF TEMPERATURE% MOD 50=0 THEN Z=YRANGE/25 ELSE Z=YRANGE/50
1420 LINE(X,Y2-Z)-STEP(0,Z):LINE(X,Y1)-STEP(0,Z)
1430 NEXT
1440 FOR Y%=2 TO 4
1450 LINE(X2,Y%)-STEP(XRANGE/50,0):LINE(X1,Y%)-STEP(-XRANGE/50,0)
1460 LOCATE (5-Y%)*6+1,3:PRINT 10^Y%;
1470 NEXT
1480 LINE(X2,FNY(X1))-(X1,FNY(X2))
1490 INC!=XRANGE/50
1500 FOR I=0 TO 49:X=X2+I*INC!
1510 LINE(X1+X2-X,FNYC(X))-(X1+X2-(X+INC!),FNYC(X+INC!)),,,&HFOOF
1520 NEXT
1530 FOR I=1 TO K
1540 CIRCLE(X1+X2-X(I),MEAN#(I)),XRANGE/250
1550 NEXT
1560 LOCATE 3,40:PRINT"Press any key for individual points":WHILE INKEY$="":WEND
1570 LOCATE 3,40:PRINT"
1580 FOR I=1 TO K:FOR J=1 TO M
1590 CIRCLE(X1+X2-X(I),Y(I,J)),XRANGE/640
1600 NEXT:NEXT
1610 LOCATE 3,40
1620 RETURN 2420
1630 'input data editing
1640 '*****
1650 LOCATE 22,1:PRINT "Move cursor to value to be changed and press return":LOC
ATE 1,18,1
1660 J=0:A$="":WHILE A$<>CHR$(13):A$="":WHILE A$="":A$=INKEY$:A$=FNAS(A$):WEND
1670 IF A$=CHR$(30) THEN J=J-1 ELSE IF A$=CHR$(31) THEN J=J+1
1680 IF J<0 THEN J=0 ELSE IF J>M+1 THEN J=M+1
1690 IF J<10 THEN X=25:Y=J+1 ELSE X=40:Y=J-7
1700 LOCATE Y,X
1710 WEND:PRINT"";
1720 IF J<>1 THEN LOCATE 23,1:INPUT"Enter new value ";NEW.VALUE
1730 IF J=0 THEN TEMPERATURE(I)=NEW.VALUE ELSE IF J>1 THEN TIME(I,J-1)=NEW.VALUE
1740 IF J=1 THEN IF N(I)<L THEN N(I)=L ELSE N(I)=L-1
1750 GOTO 430
1760 'record data to disc file
1770 '*****
1780 CLS:INPUT "Enter file name ",FILENAMES
1790 OPEN "o",1,FILENAMES
1800 PRINT#1,K:PRINT#1,L
1810 FOR I=1 TO K
1820 PRINT#1,TEMPERATURE(I):PRINT#1,N(I)
1830 FOR J=1 TO M
1840 PRINT#1,TIME(I,J)
1850 NEXT:NEXT:CLOSE#1
1860 RETURN 2420
1870 'recall data from disc file
1880 '*****
1890 CLS
1900 INPUT"Enter file name ",FILENAMES
1910 IF INSTR(FILENAMES, ".") THEN PRINT"Please use a file name without extension
":GOTO 1900
1920 OPEN "i",1,FILENAMES
1930 INPUT#1,K:INPUT#1,L
1940 IF L MOD 2 THEN M=(L+1)/2 ELSE M=L/2+1

```

```

1950 DIM N(K), TEMPERATURE(K), X(K), S1(K), SY(K), SSS(K), TIME(K,M), Y(K,M), B#
      (5), N1(3), GA(3), GB(3), GM(3), GE(3), MEAN#(K)
1960 FOR I=1 TO K
1970 INPUT#1, TEMPERATURE(I): INPUT#1, N(I)
1980 FOR J=1 TO M
1990 INPUT#1, TIME(I,J)
2000 NEXT: NEXT: CLOSE#1
2010 GOTO 420
2020 'Temperature estimate
2030 '*****
2040 CLS
2050 INPUT "Enter the time value      ", TIME
2060 TEMP.EST=FNTI(TIME): TEMP.CON=FNTEMPERATURE(FNXC(TIME)): HIC=FNTI(TIME/2) FNTI
      (TIME)
2070 PRINT: PRINT: PRINT "Temperature estimate"; TAB(25): PRINT USING "###.#"; TEMP.E
      ST
2080 PRINT "95% confidence limit"; TAB(25): PRINT USING "###.#"; TEMP.CON
2090 PRINT "Halving interval"; TAB(25): PRINT USING "###.#"; HIC
2100 PRINT: IF (TEMP.EST-TEMP.CON) >.6*HIC THEN PRINT "Extrapolation not fully jus
      tified"
2110 RETURN 2420
2120 'report to printer results according to statistical tests
2130 '*****
2140 IF FLAG(1) OR FLAG(2) OR FLAG(5) OR (FLAG(3) AND FLAG(4)) THEN GOSUB 2200 :
      RETURN
2150 IF FLAG(3) THEN PRINT#2, "Minor non-linearity"
2160 IF FLAG(4) THEN PRINT#2, "Slightly excessive dispersion, corrected": RESULT=
      TC+.6*HIC ELSE RESULT=TI
2170 PRINT#2, "TI ( HIC ) : ";: PRINT#2, USING "###.#"; RESULT;: PRINT#2, "(";: PRINT#
      2, USING "###.#"; HIC;: PRINT#2, ")";: PRINT#2, ""
2180 PRINT#2, "Confidence limit = ";: PRINT#2, USING "###.#"; TC
2190 RETURN
2200 PRINT#2, "Full TI cannot be reported."
2210 IF FLAG(1) THEN PRINT#2, "Extrapolation = ";: PRINT#2, USING "###.#"; EXTRAPOLA
      TION
2220 PRINT#2, "": IF FLAG(2) THEN PRINT#2, "": PRINT#2, "Longest mean time to end poi
      nt=";: PRIN
      T#2, USING "####"; LONGEST.MEAN.TIME
2230 IF FLAG(3) THEN PRINT#2, "": PRINT#2, "Failed F-test : F = ";: PRINT#2, USING "
      ####.#"; F;: PRINT#2, " , FO = ";: PRINT#2, USING "###.#"; FO
2240 IF FLAG(5) THEN PRINT#2, "": PRINT#2, "Excessive dispersion."
2250 PRINT#2, "": PRINT#2, "": PRINT#2, "TI = ";: PRINT#2, USING "###.#"; TI: PRINT#2, "HI
      C = ";: PRINT#2, USING "###.#"; HIC
2260 RETURN
2270 'Time estimate
2280 '*****
2290 CLS
2300 INPUT "Enter the temperature value      ", TEMPERATURE
2310 TIME.EST = CSNG(10^(FNY(FNX1(TEMPERATURE))))
2320 TIME.CON = CSNG(10^(FNYC(FNX1(TEMPERATURE))))
2330 TEMP.CON = FNTEMPERATURE(FNXC(TIME.EST))
2340 HIC=FNTI(TIME.EST/2)-TEMPERATURE
2350 PRINT: PRINT "Time estimate "; TAB(25) :PRINT USING "###.###^^^"; TIME.EST
2360 PRINT "95% confidence limit"; TAB(25) :PRINT USING "###.###^^^"; TIME.CON
2370 PRINT: PRINT: IF (TEMPERATURE-TEMP.CON)>.6*HIC THEN PRINT "Extrapolation not
      fully justified"
2380 RETURN 2420
2390 'menu of further procedures
2400 '*****
2410 LIN=1
2420 PRINT "Press any key for menu"
2430 WHILE INKEY$="" :WEND

```

```

2440 SCREEN 0,0,0:CLS:PRINT"      New data set"
2450 PRINT"      Record to disc file"
2460 PRINT"      Short report"
2470 PRINT"      Thermal endurance graph"
2480 PRINT"      Full report"
2490 PRINT"      Short report to printer"
2500 PRINT"      Full report to printer"
2510 PRINT"      Calculate temperature estimate"
2520 PRINT"      Calculate time estimate"
2530 PRINT"      Return to BASIC"
2540 LOCATE 22,1:PRINT"Move cursor to choice and press return"
2550 LOCATE 1,1,1
2560 A$="":WHILE A$<>CHR$(13):A$="":WHILE A$="":A$=FNA$(INKEY$):WEND
2570 Y=CSRLIN
2580 IF A$=CHR$(30) OR A$=CHR$(31) THEN PRINT A$;
2590 IF A$=CHR$(30) AND (Y=1) THEN LOCATE 10,1,1
2600 IF A$=CHR$(31) AND (Y=10) THEN LOCATE 1,1,1
2610 WEND
2620 ON Y GOSUB 2660,1780,3230,1350,3280,2840,2930,2040,2270
2630 CLS:END
2640 'New data set
2650 '*****
2660 CLEAR:CLS:RESTORE 60:READ WD%,HT%:GOTO 60
2670 'Calculation of probability of chi-squared
2680 '*****
2690 DATA .2316419,.31938153,-.356563782,1.781477937,-1.821255978,1.330274429
2700 RESTORE 2690:FOR I=0 TO 5:READ B#(I):NEXT
2710 PI = 4*ATN(1#)
2720 X=SQR(CHI.SQUARED#):Z=EXP(-CHI.SQUARED#/2):SUM=0
2730 IF (K-1)/2<>INT((K-1)/2) THEN GOSUB 2780 ELSE GOSUB 2750
2740 RETURN
2750 U=1:FOR I=1 TO (K-1)/2
2760 SUM=SUM + U:U = U*CHI.SQUARED#/I/2
2770 NEXT I:P = SUM*Z:RETURN
2780 U = 1/(1 + X*B#(0)):IN = 0:SUM = 0
2790 FOR I = 1 TO 5:IN = IN +B#(I)*U^I:NEXT I
2800 FOR I = 1 TO (K-2)/2:SUM = SUM+X:X = X*CHI.SQUARED#/(1+2*I):NEXT I
2810 P = SQR(2/PI)*Z*(SUM+IN):RETURN
2820 'Short report to printer
2830 '*****
2840 IF FILENAME$="" THEN PRINT"Please record data to disc first  ":RETURN 2420
2850 OPEN"O",2,FILENAME$ + ".srep"
2860 L$=STRING$(3,10):CLS:PRINT"Sending short report to ";FILENAME$;".sre file"
2870 PRINT#2,FILENAME$;L$
2880 TIME=20000!:GOSUB 1080
2890 GOSUB 2140:PRINT#2,CHR$(12)
2900 CLOSE#2:RETURN 2420
2910 'Full report to printer
2920 '*****
2930 IF FILENAME$="" THEN PRINT"Please record data to disc first  ":RETURN 2420
2940 OPEN"O",2,FILENAME$ + ".rep"
2950 L$=STRING$(3,10):CLS:PRINT"Sending report to ";FILENAME$;".rep file"
2960 TIME =20000!:GOSUB 1080
2970 PRINT#2, FILENAME$;L$
2980 GOSUB 2140:PRINT#2, CHR$(12)
2990 PRINT#2,"F (n/d) =";:PRINT#2, USING"###.####";F;:PRINT#2, "(";K-2;",";K*
(K-1);")"
3000 PRINT#2,"F0      = ";:PRINT#2, USING"###.##";F0
3010 PRINT#2,"":PRINT#2,"Chisquared =";:PRINT#2, USING"####.##";CHI.SQUARED#;:PRI
NT#2, "(";K-1;")"
3020 PRINT#2,"Probability ";:PRINT#2, USING"###.####";P;:PRINT#2, L$
3030 PRINT#2,"Temperature";TAB(25);"Mean time";TAB(50);"Variance (log)"

```

```

3040 PRINT#2,"":FOR I=1 TO K
3050 PRINT#2, TEMPERATURE(I);:PRINT#2, TAB(25);USING"####.#";10^(SY(I)/M);:PRINT
#2,TAB(500);USING"####^ ^ ^ ^";S1(I):PRINT#2,""
3060 NEXT:PRINT#2,""
3070 PRINT#2,"Pooled variance (s1)";:PRINT#2, TAB(50);USING"####^ ^ ^ ^";S1:PRINT#
2, L$
3080 FOR I=1 TO K:PRINT#2, LIN$(I):NEXT:PRINT#2, CHR$(12)
3090 GOSUB 3130
3100 CLOSE#2:RETURN 2420
3110 'Print times to failure
3120 '*****
3130 PRINT#2, FILENAME$;L$
3140 FOR I=1 TO K:PRINT#2, "Temperature ";TEMPERATURE(I)
3150 PRINT#2,"":PRINT#2,"Times:"
3160 FOR J=1 TO M
3170 PRINT#2, TAB((8*(J-1)+1)MOD 80); USING "#####.#";TIME(I,J);
3180 NEXT:PRINT#2,"":PRINT#2,"":NEXT
3190 PRINT#2, CHR$(12)
3200 RETURN
3210 'Short report to screen
3220 '*****
3230 TIME=20000!:GOSUB 1080
3240 GOSUB 1200
3250 RETURN 2420
3260 'Full report to screen
3270 '*****
3280 TIME=20000!:GOSUB 1080
3290 GOSUB 1200:PRINT"Press any key"
3300 A$="":WHILE INKEY$="":WEND:CLS
3310 PRINT"F (n,d) =";:PRINT USING"####.#";F;:PRINT "(";K-2;",";K*(M-1);")"
3320 PRINT"FO = ";:PRINT USING"###.#";FO
3330 PRINT:PRINT"Chisquared =";:PRINT USING"####.#";CHI.SQUARED#;:PRINT "(";
K-1;")"
3340 PRINT"Probability ";:PRINT USING"###.^ ^ ^ ^";P:PRINT
3350 PRINT"Temperature";TAB(25);"Mean time";TAB(50);"Variance (log)"
3360 PRINT:FOR I=1 TO K
3370 PRINT TEMPERATURE(I);:PRINT TAB(25);USING"####.#";10^MEAN#(I);:PRINT TAB
(50);USING"####^ ^ ^ ^";S1(I)
3380 NEXT:PRINT
3390 PRINT"Pooled variance (s1)";:PRINT TAB(50);USING"####^ ^ ^ ^";S1:PRINT
3400 RETURN 2420
3410 DATA 4, 5, .369332835 , -.706772779D-01, .427490560 ,.218572320
3420 DATA 4, 6, .395429409 , -.582745952D-01, .222691480 ,.208957080
3430 DATA 4, 7, .415588035 , -.465401553D-01, 0 ,.210446860
3440 DATA 5, 7, .289187215 , -.380049078D-01, .364264390 ,.167482900
3450 DATA 5, 8, .302258274 , -.320461311D-01, .173745210 ,.164630210
3460 DATA 5, 9, .312981200 , -.263842700D-01, 0 ,.166101280
3470 DATA 6, 9, .236385800 , -.232986100D-01, .296052680 ,.136888360
3480 DATA 6,10, .244118092 , -.200045322D-01, .142374020 ,.135971440
3490 DATA 6,11, .250685932 , -.168530354D-01, 0 ,.137162430
3500 DATA 7,11, .199470571 , -.155838805D-01, .249259910 ,.116072110
3510 DATA 7,12, .204533441 , -.135763751D-01, .120575110 ,.115854770
3520 DATA 7,13, .208940612 , -.116456143D-01, 0 ,.116799000
3530 DATA 8,13, .172347188 , -.110866900D-01, .215202200 ,.100909060
3540 DATA 8,14, .175900510 , -.977439669D-02, .104554720 ,.100938940
3550 DATA 8,15, .179051341 , -.850715308D-02, 0 ,.101694650
3560 DATA 9,15, .151627832 , -.825677607D-02, .189315610 ,.892954300D-01
3570 DATA 9,16, .154250858 , -.735251258D-02, .922870600D-01, .894321900D-01
3580 DATA 9,17, .156610476 , -.647646027D-02, 0 ,.900465800D-01
3590 DATA 10,17, .135307112 , -.636989298D-02, .168979490 ,.801067000D-01
3600 DATA 10,18, .137318956 , -.572067003D-02, .825931100D-01, .802834400D-01
3610 DATA 10,19, .139149625 , -.509001812D-02, 0 ,.807909800D-01

```

```

3620 DATA 11,19, .122130237 , -.505348094D-02, .152583850 ,.726473000D-01
3630 DATA 11,20, .123720203 , -.457177272D-02, .747405200D-01,.728342900D-01
3640 DATA 11,21, .125180504 , -.410278708D-02, 0 ,.732597474D-01
3650 DATA 12,21, .111274798 , -.410102462D-02, .139085629 ,.664671650D-01
3660 DATA 12,22, .112561882 , -.373384010D-02, .682500019D-01,.666514225D-01
3670 DATA 12,23, .113753115 , -.337566146D-02, 0 ,.670122916D-01
3680 DATA 13,23, .102180512 , -.339101130D-02, .127779807 ,.612611371D-01
3690 DATA 13,24, .103243093 , -.310475468D-02, .627963988D-01,.614362434D-01
3700 DATA 13,25, .104232886 , -.282505015D-02, 0 ,.617462540D-01
3710 DATA 14,25, .944530651D-01, -.284836965D-02,.118172726 ,.568145550D-01
3720 DATA 14,26, .953447768D-01, -.262091053D-02,.581494476D-01,.569782472D-01
3730 DATA 14,27, .961799524D-01, -.239833079D-02,0 ,.572472927D-01
3740 DATA 15,27, .878071666D-01, -.242479697D-02,.109908494 ,.529718566D-01
3750 DATA 15,28, .885659092D-01, -.224107889D-02,.541425782D-01,.531236252D-01
3760 DATA 15,29, .892798839D-01, -.206106835D-02,0 ,.533592291D-01
3770 DATA 16,29, .820314575D-01, -.208812750D-02,.102724091 ,.496174290D-01
3780 DATA 16,30, .826847462D-01, -.193761963D-02,.506521117D-01,.497576046D-01
3790 DATA 16,31, .833019925D-01, -.178997874D-02,0 ,.499655775D-01
3800 DATA 1

```

부속서 E

불완전한 데이터의 그룹평균, 분산 및 분산평균 계수

n	m	μ	α	β	ε
10	6	0,14237402	0,244118092	-2,00045322E-2	0,13597144
11	6	0	0,250685932	-1,68530354E-2	0,13716243
11	7	0,24925991	0,199470571	-1,55839905E-2	0,11607211
12	7	0,12057511	0,204533441	-1,35763751E-2	0,11585477
13	7	0	0,208940612	-1,16456143E-2	0,116799
13	8	0,2152022	0,172347188	-1,108669E-2	0,10090906
14	8	0,10455472	0,17590051	-9,77439669E-3	0,10093894
15	8	0	0,179051341	-8,50715308E-3	0,10169465
15	9	0,18931561	0,151627832	-8,25677607E-3	8,929543E-2
16	9	9,228706E-2	0,154250858	-7,35251258E-3	8,943219E-2
17	9	0	0,156610476	-6,47646027E-3	9,004658E-2
17	10	0,16897949	0,135307112	-6,36989298E-3	8,01067E-2
18	10	8,259311E-2	0,137318956	-5,72067003E-3	8,028344E-2
19	10	0	0,139149625	-5,09001812E-3	8,079098E-2
19	11	0,15258385	0,122130237	-5,05348094E-3	7,26473E-2
20	11	7,474052E-2	0,123720203	-4,57177272E-3	7,283429E-2
21	11	0	0,125180504	-4,10278708E-3	7,32597474E-2
21	12	0,139085629	0,111274798	-4,10102462E-3	6,6467165E-2
22	12	6,82500019E-2	0,112561882	-3,7338401E-3	6,66514225E-2
23	12	0	0,113753115	-3,37566146E-3	6,70122916E-2
23	13	0,127779807	0,102180512	-3,3910113E-3	6,12611371E-2
24	13	6,27963988E-2	0,103243093	-3,10475468E-3	6,14362434E-2
25	13	0	0,104232886	-2,82505015E-3	6,1746254E-2
25	14	0,118172726	9,44530651E-2	-2,84836965E-3	5,6814555E-2
26	14	5,81494476E-2	9,53447768E-2	-2,62091053E-3	5,69782472E-2
27	14	0	9,61799524E-2	-2,39833079E-3	5,72472927E-2
27	15	0,109908494	8,78071666E-2	-2,42479697E-3	5,29718566E-2
28	15	5,41425782E-2	8,85659092E-2	-2,24107889E-3	5,31236252E-2
29	15	0	8,92798839E-2	-2,06106835E-3	5,33592291E-2
29	16	0,102724091	8,20314575E-2	-2,0881275E-3	4,9617429E-2
30	16	5,06521117E-2	8,26847462E-2	-1,93761963E-3	4,97576046E-2
31	16	0	8,33019925E-2	-1,78997874E-3	4,99655775E-2
31	17	9,64209444E-2	7,69660991E-2	-1,81630106E-3	4,66635039E-2

부속서 F

첫 번째 사이클 내의 끝점 도달 단일시간 및 복수시간

F.1 첫 번째 사이클 내의 끝점 도달 단일시간

끝점이 첫 번째 사이클에서 하나의 시료에 도달하는 경우, 이는 비정상 환경(예 : 고장난 시료) 때문일 가능성이 있다. 수학적으로 시간 0과 1(1사이클 단위) 사이의 끝점 도달시간이 $y = \log(\text{시간})$ 의 값으로 할당될 수는 없다.

다음과 같은 두 개의 대안 절차가 있다.

i) 이 그룹을 버리고 다른 그룹을 재검정한다.

ii) 시험을 계속하고, 끝점도달 m 시간이 기록될 때까지 끝점도달 첫 번째 시간(무효한 것으로 가정)을 무시한다.

여기에서 $m = (n + 1)/2$ n 이 홀수인 경우

$m = n/2 + 1$ n 이 짝수인 경우

n 은 각 그룹에서 시료의 원래의 수이다.

부록 E의 검사는 $\varepsilon_{n,m}$ 이 $\varepsilon_{n-1, m}$ 에 매우 가깝다는 것을 보여준다. 따라서, 산출은 모든 그룹과 μ , α , β 의 해당 값으로 실시할 수 있다.

F.2 첫 번째 사이클 내의 끝점도달 복수 시간

끝점이 첫 번째 사이클에서 하나 이상의 시료에 도달한다면, 해당 그룹에서 얻은 데이터를 사용하여 분석할 정당화하는 것은 불가능하다. 가장 합리적인 설명은 결함 기법, 사이클 기간 선택 또는 실험 기법(샘플 준비 또는 처리)이다. 이 그룹을 버리고 해당 온도에 대한 시험을 권장한다.